
Consideremos un cilindro circular de radio R hecho de algún material con

respuesta dieléctrica ε sobre el que incide una onda plana monocromática de

frecuencia ω. Orientemos el eje z a lo largo del eje del cilindro y el eje x a lo

largo de la dirección de propagación, como en la �gura 1.

1. Caso transversal magnético (TM)

1.1. Campo incidente

Consideremos primero el caso en que el campo magnético de la onda

incidente Hi apunta en la dirección z. Por simetría, el campo magnético de

todas las ondas esparcidas también apuntaría en la dirección z. El campo

eléctrico estaría en el plano xy. El campo incidente puede escribirse como

Hi = Hi0ẑe
i(qx−ωt), (1)

donde q = ω/c y c es la velocidad de la luz. Normalizamos la amplitud del

campo incidente,Hi0 = 1, ignoramos la dependencia temporal de los campos,

pues todos tienen la exponencial e−iωt, la cual puede cancelarse, y expresa-

mos eiqx = eiqr cos θ en términos de funciones de Bessel y eigenfunciones del

ímpetu angular L̂z
eiqx =

∑
l

ilJl(qr)e
ilθ, (2)

donde r y θ son las coordenadas polares radial y angular, l es el eigenvalor
del ímpetu angular y Jl son las funciones de Bessel del primer tipo.

1.2. Campo en el exterior

En el vacío la componente z del campo magético obedece la ecuación

de onda, que para ondas monocromáticas se convierte en la ecuación de

1



~q

z

x

y

R

Figura 1: Sistema consistente en un cilindro de radio R sobre el que incide

una onda plana con vector de onda q propagándose en la dirección x, y
produciendo ondas esparcidas.

Helmholtz,

(∇2 + q2)H = 0. (3)

En coordenadas cilíndricas, el laplaciano puede escribirse como

∇2 =
1

r

∂

∂r
r
∂

∂r
+

1

r2
∂2

∂θ2
, (4)

y la ecuación de Helmholtz se puede resolver por separación de variables

H(r, θ) = f(r)g(θ). La solución de la parte angular con condiciones perió-

dicas $g(θ)=g(θ+2π) $es de la forma g(θ) = eilθ con l entero, y la ecuación

correspondiente para la parte radial es

1

r

∂

∂r
r
∂

∂r
f(r)− l2

r2
f(r) + q2f(r) = 0, (5)

que se puede reescribir como

χ2f ′′(χ) + χf ′(χ) + (χ2 − l2)f(χ) = 0, (6)

donde hicimos el cambio de variable χ ≡ qr y donde f ′′ y f ′ denotan las

derivadas de f respecto a su argumento. Identi�camos esta como la ecuación

de Bessel y sus soluciones son alguna combinación de ondas de Bessel corres-

pondientes al momento angular l. Como era de esperarse, cada uno de los

sumandos en la expansión (2) satisface la ec. de onda. El campo producido

por el cilíndro debe consistir de ondas salientes en el límite r → ∞, y por

tanto el campo total en el vacío debe ser de la forma

H(r, θ) =
∑
l

il(Jl(qr) + alH
(1)
l (qr))eilθ, r > R (7)

donde H
(1)
l (qr)) son las funciones de Hankel salientes y al son coe�cientes a

determinar.

1.3. Campo en el interior

En el interior del cilindro también se cumple una ecuación de onda aná-

loga a la ec. (3) pero sustituyendo q por el número de onda k dentro del

material, donde

k2 = εq2. (8)

La solución en el interior debe ser regular en el origen, y por lo tanto puede

escribirse como

H(r, θ) =
∑
l

ilblJl(kr)e
ilθ, r < R (9)

donde bl son coe�cientes a determinar.

3



1.4. Condiciones de frontera

Para obtener los coe�cientes al y bl recurrimos a las condiciones de fron-

tera en la super�cie r = R. El campo magnético es tangencial a la super�cie

del cilindro y por tanto debe ser contínuo, con el mismo valor en R+ que en

R−. Por tanto∑
l

il(Jl(qR) + alH
(1)
l (qR))eilθ =

∑
l

ilblJl(kR)e
ilθ. (10)

Como las funciones trigonométricas eilθ forman una base ortogonal del espa-

cio de funciones angulares y son linealmente independientes, podemos elimi-

narlas junto con la suma en la ec. (10) y simpli�car la condición de frontera

escribiéndola como

(Jl(qR) + alH
(1)
l (qR)) = blJl(kR). (11)

Empleando la ec. de Ampére-Maxwell podemos hallar el desplazamiento eléc-

trico D a partir del campo magnético H. La componente tangencial del

desplazamiento es entonces

Dθ =
1

q

∂

∂r
H. (12)

Como la componente tangencial del campo eléctrico es continua en R, y
E = D en el exterior y E = D/ε en el interior del material, entonces

εq(J ′l (qR) + al(H
(1))′l(qR)) = kblJ

′
l (kR). (13)

1.5. Esparcimiento

Podemos eliminer bl de las ecuaciones (11) y (13) para obtener �nalmente

los coe�cientes de esparcimiento al

al = −
εqJ ′l (qR)Jl(kR)− kJl(qR)J ′l (kR)

εq(H(1))′l(qR)Jl(kR)− kH
(1)
l (qR)J ′l (kR)

(14)

2. Caso transversal eléctrico (TE)

En este caso el campo eléctrico apunta en la dirección z y en analogía

con el caso TM, escribimos

E(r, θ) =
∑
l

il(Jl(qr) + alH
(1)
l (qr))eilθ, r > R (15)

E(r, θ) =
∑
l

ilblJl(kr)e
ilθ, r < R (16)

en analogía a las ecs. (7) y (9). El magnético paralelo puede obtenerse de la

ley de Faraday, por lo que

Bθ =
1

q

∂

∂r
E. (17)

Tanto E como Bθ son contínuos en R, lo cual lleva a las condiciones de

contorno

(Jl(qR) + alH
(1)
l (qR)) = blJl(kR). (18)

y

q(J ′l (qR) + al(H
(1))′l(qR)) = kblJ

′
l (kR). (19)

Resolviendo, podemos obtener las amplitudes de dispersión

al = −
qJ ′l (qR)Jl(kR)− kJl(qR)J ′l (kR)

q(H(1))′l(qR)Jl(kR)− kH
(1)
l (qR)J ′l (kR)

(20)

3. Secciones transversales

Para distancias grandes, las funciones de Bessel pueden aproximarse por

H
(1)
l (qr) =

√
2

πqr
ei(qr−(l+1/2)π/2), (21)

Por lo tanto, el �ujo de energía esparcida es

S =
c

4π2qr
Re
∑
ll′

ala
∗
l′e
i(l−l′)θe−i(l−l

′)π/2 (22)

Multiplicando por r obtenemos la potencia esparcida por unidad de longitud

(axial) y por unidad de ángulo, y dividiendo entre la intensidad incidente

Ii =
c

8π
, (23)

obtenemos la sección diferencial de esparcimiento

dσes
dθ

=
2

πq
Re
∑
ll′

ala
∗
l′e
i(l−l′)θ (24)

Como el problema es 1D esta sección tiene unidades de longitud, no de área.

Integrando sobre el ángulo obtenemos la sección total de esparcimiento.

σes =
4

q

∑
l

|al|2. (25)

En la dirección frontal hay interferencia entre el campo incidente y el

campo esparcido. Escribimos el campo total F

F (r) = eiqx +
f(θ)eiqr√

r
, (26)

donde identi�camos a F con el campo magnético para el caso TM y con el

campo eléctrico para el caso TE, donde

f(θ) =

√
2

πq

∑
l

ilale
ilθe−i(l+1/2)π/2 =

√
2

iπq

∑
l

ale
ilθ (27)

Para ángulos pequeños hacemos las aproximaciones

f(θ) ≈ f(0) =
√

2

iπq

∑
l

al, (28)

eiqr ≈ eiqxeiqy
2/2x. Si colocamos un detector en la dirección frontal, la po-

tencia por unidad de longitud que recibiría sería

P =
1

8π

(
`+ 2Re

f(0)√
x

∫ `/2

−`/2
dy eiqy

2/2x

)
+ . . . (29)

donde ` es la apertura del detector y el término que despreciamos es del

orden de 1/
√
qr veces menor que el que conservamos. Suponiendo que el

detector cubre un ángulo pequeño, `� x, pero que es tan ancho que eiqy
2/x

tiene muchas oscilaciones, en su interior, i.e. ell2 � λx, entonces

P =
c

8π

(
`+ 2Re f(0)

√
2πi

q

)
, (30)

donde usamos la integral ∫ ∞
−∞

dζ eiζ
2
=
√
iπ, (31)

como se puede veri�camos en el apéndice 8. El primer término en la ec. (30)

es la potencia que llegaría al detector de no haber partículo. Por lo tanto, el

cociente del segundo al primero puede identi�carse en términos de la fracción

de energía que no llega al detector debido a la presencia de la partícula, i.e.,

a la sección transversal de extinción,

σex = −2Re f(0)

√
2πi

q
= −4

q
Re
∑
l

al. (32)

La sección transversal de absorción está dada por la diferencia

σabs = σex − σes. (33)

4. Campo super�cial.

Para evaluar el campo dentro del cilindro debemos obtener los coe�cientes

bl. Empleando las ecs. (11) o (18) obtenemos para ambas polarizaciones

bl =
Jl(qR) + alH

(1)
l (qR)

Jl(kR)
. (34)

El campo eléctrico tangencial super�cial puede evaluarse a partir de las

ecs. (12) para polarización TM y de (16) para polarización TE usando con-

diciones de continuidad. Podemos obtener una medida de su magnitud pro-

mediando su cuadrado sobre la super�cie,

〈E2
θ 〉 =

∣∣∣∣kε
∣∣∣∣2∑

l

|blJ ′l (kR)|2, TM (35)

〈E2〉 =
∑
l

|blJl(kR)|2. TE (36)

5. Límite no retardado

En el caso no retardado qR � 1 puedo obtener resultados aproximados

tomando los valores asintóticos de las funciones de Bessel,

Jl(ζ) ≈
ζ l

2ll!

N0(ζ) ≈
2

π
log ζ

Nl(ζ) ≈ −
(l − 1)!

π

(
2

ζ

)l
, l > 0

(37)

Sustituyendo en la ec. (14) obtenemos para l > 1

al = iπ(qR)2l
l

(2ll!)2
ε− 1

ε+ 1
. (38)

Multiplicando por la función de Hankel correspondiente, obtenemos el campo

magnético inducido en el exterior, dado por

H ind(r) =
∑
l

(iq)l
ε− 1

ε+ 1

1

2ll!

R2l

rl
eilθ ≈ iq ε− 1

ε+ 1
R2 x

r2
, (39)

dominado por las contribuciones l = ±1. Aquí he ignorado el término con

l=0 pues no contribuye al campo eléctrico. El campo eléctrico inducido puede

obtenerse de la ec. de Ampére-Faraday

Eind = − 1

iq
∇×H ind = −ε− 1

ε+ 1
R2

(
∂

∂y

x

r2
,
∂

∂x

x

r2

)
. (40)

Notamos que x/r2 = ∂ log r/∂x y que ∂2 log r/∂x2 = −∂2 log r/∂y2, pues
log r obedece la ec. de Laplace, por lo que podemos reescribir la ec. (40)

como

Eind = −2p · ∇∇ log r = (2p · rr − pr2)/r4, (41)

donde identi�camos el dipolo inducido por unidad de longitud

p = αEex, (42)

donde el campo eléctrico externo tiene la misma magnitud que el campo

magnético externo (que hemos normalizado a 1) y apunta en la dirección y,
y donde identi�camos la polarizabilidad por unidad de longitud

α =
ε− 1

ε+ 1

R2

2
. (43)

Este es un resultado bien conocido que puede obtenerse mediante un desa-

rrollo multipolar cuasiestático.

6. Código

6.1. Dispersión

A continuación presento un código para el cálculo de la dispersión.

Empiezo el código cargando los paquetes y las opciones usuales.

use v5.12;

use warnings;

use List::Util;

use Getopt::Long;

use IO::Prompter;

use PDL;

use PDL::NiceSlice;

use PDL::Graphics::Gnuplot;

Figura 2: paquetes

Declara e inicializa los parámetros.

my $eps; # función dieléctrica del cilindro.

my $R=1; # Usa el radio como unidad de distancia.

my $c=1; # R/c como unidad de tiempo.

my ($wmin, $wmax, $Nw); # frecuencias mínima, máxima y número

my $lmax; # maximo momento angular

my $options=q(

'eps=f'=>\$eps, # función dieléctrica del cilindro

'wmin=f'=>\$wmin, # frecuencia mínima

'wmax=f'=>\$wmax, # frecuencia máxima

'Nw=i'=>\$Nw, #número de frecuencias

'lmax=i'=>\$lmax, # máximo momento angular

);

my %options=(eval $options);

die "Error en definición de opciones; $@" if $@;

GetOptions(%options) or usage($options, "Opciones erróneas");

usage($options, "Faltan parámetros")

unless List::Util::all {defined $_} ($eps, $wmin, $wmax, $Nw, $lmax);

my $w=zeroes($Nw)->xlinvals($wmin,$wmax); # Arreglo de frecuencias

my $qR=$w/$c*$R;

my $kR=sqrt($eps)*$w*$R;

my $l=sequence($lmax+1);

Figura 3: inicializa

Necesitamos una rutina para explicar el uso si hubiera un error.

sub usage {

my ($options, $message)=@_;

say $message if defined $message;

say $options;

exit 1;

}

Figura 4: uso

Calcula

# Calcula las funciones de Bessel, Neumann y Hankel

my $JlqR=bessjn($qR->dummy(1), $l->dummy(0));

my $NlqR=bessyn($qR->dummy(1), $l->dummy(0));

my $HlqR=$JlqR+i()*$NlqR;

my $JlkR=bessjn($kR->dummy(1), $l->dummy(0));

# Calcula sus derivadas. Itera sobre frecuencias. Pon l como 1er índice.

my $DJlqR=deriva($JlqR->mv(1,0), $qR->dummy(0))->mv(0,1);

my $DHlqR=deriva($HlqR->mv(1,0), $qR->dummy(0))->mv(0,1);

my $DJlkR=deriva($JlkR->mv(1,0), $kR->dummy(0))->mv(0,1);

# factor 1,2,2,2... para contar l's negativas

my $factor_l=(1+($l>=1))->dummy(0);

# Polarización TM

# Coeficientes

my $al_TM=-($eps*$qR*$DJlqR*$JlkR-$kR*$JlqR*$DJlkR)

/($eps*$qR*$DHlqR*$JlkR-$kR*$HlqR*$DJlkR);

my $bl_TM=($JlqR+$al_TM*$HlqR)/$JlkR;

# Secciones

my $sigma_s_TM=($factor_l*4/$qR*$al_TM->abs2)->mv(1,0)->sumover;

my $sigma_ex_TM=-($factor_l*4/$qR*$al_TM->re)->mv(1,0)->sumover;

# Campo

my $E2_TM=($factor_l*($kR/$eps*$bl_TM*$DJlkR)->abs2)->mv(1,0)->sumover;

# Polarización TE

# Coeficientes

my $al_TE=-($qR*$DJlqR*$JlkR-$kR*$JlqR*$DJlkR)

/($qR*$DHlqR*$JlkR-$kR*$HlqR*$DJlkR);

my $bl_TE=($JlqR+$al_TE*$HlqR)/$JlkR;

# Secciones

my $sigma_s_TE=($factor_l*4/$qR*$al_TE->abs2)->mv(1,0)->sumover;

my $sigma_ex_TE=-($factor_l*4/$qR*$al_TE->re)->mv(1,0)->sumover;

# Campo

my $E2_TE=($factor_l*($bl_TE*$JlkR)->abs2)->mv(1,0)->sumover;

Figura 5: calcula

Necesito una subrutina para derivar las funciones de Bessel. Uso la expre-

sión recursiva que relacionia la derivada de un orden con el orden anterior.

El caso l=0 es especial. Asumo que l es el primer índice. De haber más, itera

automáticamente.

sub deriva { #l es el primer índice

my ($Bl, $x) = @_; # B(x,l), B is any Bessel, $x are arguments of Bessel

my $Blm1=$Bl->rotate(1); # B(x,l-1)

my $DBl=-$Bl->xvals/$x*$Bl+$Blm1;

$DBl((0)).=-$Bl((1));

return $DBl;

}

Figura 6: deriva

Por completez, va abajo la versión en términos de la siguiente en lugar

de la previa función de Bessel. En este caso, la derivada del orden máximo

está equivocada y no debe usarse.

sub deriva {

my ($Bl, $x) = @_; # B(x,l), B is any Bessel, $x are arguments of Bessel

my $Blp1=$Bl->rotate(-1); # B(x,l+1)

my $DBl=$Bl->xvals/$x*$Bl-$Blp1; # don't trust higest l;

return $DBl;

}

Figura 7: deriva-alt

Gra�ca

my $gw=gpwin();

$gw->plot({title=>'F. Bessel del primer tipo', xlabel=>'qR', yzeroaxis=>1},

with=>'lines', $qR, $JlqR);

prompt -single, -void, "Listo?";

$gw->plot({title=>'Derivadas de Bessel del primer tipo', xlabel=>'qR', yzeroaxis=>1},

with=>'lines', $qR, $DJlqR);

prompt -single, -void, "Listo?";

$gw->plot({title=>'F. Bessel del segundo tipo', xlabel=>'qR', yzeroaxis=>1, yrange=>[-1,.5]},

with=>'lines', $qR, $NlqR, with=>'lines', $qR, pdl(0));

prompt -single, -void, "Listo?";

$gw->plot({title=>"Coeficientes de esparcimiento TM {/Symbol e}=$eps", xlabel=>'qR', yzeroaxis=>1},

with=>'lines', $qR, $al_TM->abs2); # remueve la l más alta

prompt -single, -void, "Listo?";

$gw->plot({title=>"Secciones transversales TM {/Symbol e}=$eps", xlabel=>'qR', yzeroaxis=>1},

with=>'lines', $qR, $sigma_s_TM,

with=>'lines', $qR, $sigma_ex_TM,

);

prompt -single, -void, "Listo?";

$gw->plot({title=>"Campo cuadrado promedio TM {/Symbol e}=$eps", xlabel=>'qR', yzeroaxis=>1},

with=>'lines', $qR, $E2_TM);

prompt -single, -void, "Listo?";

$gw->plot({title=>"Coeficientes de esparcimiento TE {/Symbol e}=$eps", xlabel=>'qR', yzeroaxis=>1},

with=>'lines', $qR, $al_TE->abs2); # remueve la l más alta

prompt -single, -void, "Listo?";

$gw->plot({title=>"Secciones transversales TE {/Symbol e}=$eps", xlabel=>'qR', yzeroaxis=>1},

with=>'lines', $qR, $sigma_s_TE,

with=>'lines', $qR, $sigma_ex_TE,

);

prompt -single, -void, "Listo?";

$gw->plot({title=>"Campo cuadrado promedio TE {/Symbol e}=$eps", xlabel=>'qR', yzeroaxis=>1},

with=>'lines', $qR, $E2_TE);

prompt -single, -void, "Listo?";

Figura 8: gra�ca

Finalmente, juntamos todo en un programa que gra�ca los espectros de

los coe�cientes de dispersión, las secciones transversales y el campo super�-

cial.

#Calcula los coeficientes de esparcimiento de un cilindro dieléctrico.

<<paquetes>>

<<inicializa>>

<<calcula>>

<<deriva>>

<<grafica>>

<<uso>>

Figura 9: todo

6.2. Resultados

Corrimos el programa 9 como en la �gura 10 indicando una función die-

léctrica ε = 10, y haciendo cálculos para 1000 frecuencias entre .1 y 3.5,

empleando momentos angulares l = 0, 1, . . . 10.

./mie2D.pl -eps 10 -wmin .1 -wmax 3.5 -Nw 1000 -lmax 10

Figura 10: Ejecución del programa 9.

La �gura 11 muestra las primeras funciones de Bessel del primer tipo.

Son regulares en el origen, donde son aproximadamente potencias de su ar-

gumento, i.e., J0 es constante, J1 lineal, J2 cuadrática y así sucesivamente.

En la �gura 12 mostramos las derivadas de las funciones de Bessel, para ve-

ri�car cualitativamente que correspondan a lo esperado a partir de la �g. 11.

En la �gura 13 mostramos las primeras funciones de Bessel del segundo tipo

de orden entero, o funciones de Neumann Nl. Observamos su singularidad

Figura 11: Primeras funciones de Bessel del primer tipo Jl con órdenes enteros
l = 0, 1, 2 . . ..

Figura 12: Derivadas J ′l respecto a su argumento de las primeras funciones

de Bessel del primer tipo.

Figura 13: Funciones de Neumann Nl con órdenes enteros l = 0, 1, 2 . . .

en el origen. Con estas podemos obtener las funciones de Hankel salientes

empleadas en el cálculo de esparcimiento.

Las �guras 14 y 15 muestra los coe�cientes de dispersión correspondien-

tes a las polarizaciones TM y TE, respectivamente correspondientes a un

cilindro dieléctrico con permitividad ε = 10 iluminado con una onda plana

monocromática de frecuencia ω = qc de amplitud 1. Notamos que el valor

máximo para los coe�cientes de dispersión es 1. Esto se debe a que distintos

momentos angulares no se mezclan debido a la simetría rotacional, y para

cada valor l del momento angular, no puede esparcirse más energía que la que

incide. Las formas de las curvas son curiosas y algunas muestran resonancias

muy delgadas mientras que otras no.

Las �guras 16 y 17 muestran el cuadrado del campo eléctrico paralelo

a la super�cie promediado sobre ella para ambas polarizaciones. Notamos

que aparecen una serie de picos angostos y altos. Estos corresponden a las

resonancias de Mie 2D. Notamos que para toda l, para distancias su�cien-

temente largas el campo esparcido se vuelve propagante. Por tanto, siempre

hay pérdida de energía de estos modos, aunque los materiales mismos sean

dieléctricos que no presenten absorción. Por tanto, las resonancias de Mie

Figura 14: Coe�cientes de dispersión al evaluados en el exterior de un ci-

lindro dieléctrico con permitividad ε = 10 como función de la frecuencia

normalizada qR.

Figura 15: Coe�cientes de dispersión como en la �g. 14 pero para polarización

TE.

Figura 16: Cuadrado del campo electrico tangencial E2
θ promediado sobre la

super�cie del cilindro para polarización TM.

Figura 17: Cuadrado del campo electrico tangencial E2
z promediado sobre la

super�cie del cilindro para polarización TE.

tienen necesariamente anchos y alturas �nitas. Los anchos disminuyen con-

forme aumenta el índice de la resonancia, hasta llegar a la resonancia situada

alrededor de qR ≈ 2,7. Para polarización TM el campo alcanza ampli�cacio-

nes mayores que para polarización TE.

Finalmente, en las �guras 18 y 19 mostramos las secciones de esparci-

miento y de extinción, σes y σex para ambas polarizaciones. Notamos que

no hay diferencia apreciable entre ellas, pues no introdujimos disipación. La

forma de las curvas se ve bastante caprichosa.

Figura 18: Secciones transversales σes y σex para polarización TM.

6.3. Campo

Ahora presento un código para la visualización de los campos. Para car-

gar los paquetes y las opciones usuales reciclo el código de la �g. 2. Los

parámetros son ahora diferentes.

my $eps; # función dieléctrica del cilindro.

my $R=1; # Usa el radio como unidad de distancia.

my $c=1; # R/c como unidad de tiempo.

my $w; # frecuencias para el cálculo

Figura 19: Secciones transversales σes y σex para polarización TM.

my $lmax; # maximo momento angular

my $xmax; # x o y máximas para graficar, en unidades del radio R

my $N; # 2N+1 es el número de pixeles en x y en y

my $Rcut=$R; # cut plot before this radius (default 1)

my $options=q(

'eps=f'=>\$eps, # función dieléctrica del cilindro

'w=f'=>\$w, # frecuencia

'lmax=i'=>\$lmax, # máximo momento angular

'xmax=f'=>\$xmax, # x o y distancia máximos para graficar, en unidades del radio R

'N=i'=>\$N, # 2N+1 es el número de pixeles en x y en y

'Rcut=f'=>\$Rcut; # cut plot before this radius

);

my %options=(eval $options);

die "Error en definición de opciones; $@" if $@;

GetOptions(%options) or usage($options, "Opciones erróneas");

usage($options, "Faltan parámetros")

unless List::Util::all {defined $_} ($eps, $w, $lmax, $xmax, $N);

my $qR=$w/$c*$R;

my $kR=sqrt($eps)*$w*$R;

my $l=sequence($lmax+1);

my $cartesian=(zeroes(2*$N+1,2*$N+1)->ndcoords-$N)/$N*$xmax; # cartesian coordinates

my $r=$cartesian->abs2->sumover->sqrt; # magnitudes

my $theta=atan2($cartesian((1)), $cartesian((0))); # angles

Figura 20: inicializa1

En caso de errores, reciclamos la rutina de la �g. 4. Calculamos los co-

e�cientes de dispersión de una manera similar a la �g. 5, pero para una sola

frecuencia.

# Calcula las funciones de Bessel, Neumann y Hankel en la superficie

my $JlqR=bessjn($qR, $l); # l

my $NlqR=bessyn($qR, $l);

my $HlqR=$JlqR+i()*$NlqR;

my $JlkR=bessjn($kR, $l);

# Calcula sus derivadas.

my $DJlqR=deriva($JlqR, $qR);

my $DHlqR=deriva($HlqR, $qR);

my $DJlkR=deriva($JlkR, $kR);

# factor 1,2,2,2... para contar l's negativas

my $factor_l=(1+($l>=1));

# Polarización TM

my $al_TM=-($eps*$qR*$DJlqR*$JlkR-$kR*$JlqR*$DJlkR)

/($eps*$qR*$DHlqR*$JlkR-$kR*$HlqR*$DJlkR);

my $bl_TM=($JlqR+$al_TM*$HlqR)/$JlkR;

# Polarización TE

my $al_TE=-($qR*$DJlqR*$JlkR-$kR*$JlqR*$DJlkR)

/($qR*$DHlqR*$JlkR-$kR*$HlqR*$DJlkR);

my $bl_TE=($JlqR+$al_TE*$HlqR)/$JlkR;

Figura 21: coe�cientes

Puedo reciclar el código de la �gura 6 para calcular las derivadas de las

funciones de Bessel. Ahora puedo calcular el campo esparcido y su cuadrado,

proporcional al �ujo de energía en la zona de radiación. Lo escalo con r para
que a largas distancias coincida con la sección diferencial de dispersión.

my $il=($l%4==0)+i()*($l%4==1)-($l%4==2)-i()*($l%4==3); #[l] i^l

my $m1l=2*($l%2==0)-1; # [l] -1^l

my $eiltheta=exp(i()*$theta->dummy(0)*$l); # [l,nx,ny]

my $Jlqr=bessjn($qR*$r->dummy(0), $l); # [l,nx,ny]

my $Nlqr=bessyn($qR*$r->dummy(0), $l); # [l,nx,ny]

my $Hlqr=$Jlqr+i()*$Nlqr; # [l,nx,ny]

# use H_{-1}=H_l, i^{-l}=(i^l)^* and e^{-il\theta}=(e^{il\theta})^*

# a_{-l}=a_l

# TM

my $H_TM=($il*$al_TM*$Hlqr*$eiltheta)->sumover +

($il->conj*$al_TM*$m1l*$Hlqr*$eiltheta->conj)->(1:-1)->sumover;

my $S_TM=$H_TM->abs2*($r>$Rcut);

# TE

my $E_TE=($il*$al_TE*$Hlqr*$eiltheta)->sumover +

($il->conj*$al_TE*$m1l*$Hlqr*$eiltheta->conj)->(1:-1)->sumover;

my $S_TE=$E_TE->abs2*($r>$Rcut);

Figura 22: poynting

Gra�ca los resultados.

my $gw=gpwin();

$gw->plot({title=>"r*H²/H²_i TM {/Symbol e}=$eps {/Symbol w}=$w ",

xlabel=>'x', ylabel=>'y', justify=>1},

with=>'image', $cartesian((0)), $cartesian((1)), $r*$S_TM);

prompt -single, -void, "Listo?";

$gw->plot({title=>"r*E^2/E_i^2 TE {/Symbol e}=$eps {/Symbol w}=$w ",

xlabel=>'x', ylabel=>'x', justify=>1},

with=>'image', $cartesian((0)), $cartesian((1)), $r*$S_TE);

prompt -single, -void, "Listo?";

Figura 23: gra�ca1

Finalmente, juntamos todo en un programa que gra�ca la intensidad

esparcida.

#Calcula los coeficientes de esparcimiento de un cilindro dieléctrico.

<<paquetes>>

<<inicializa1>>

<<coeficientes>>

<<poynting>>

<<grafica1>>

<<deriva>>

<<uso>>

Figura 24: todo1

6.4. Resultados

Corrimos el programa 24 como en la �gura 25 indicando una función

dieléctrica ε = 10, y haciendo cálculos para diversas frecuencias ωR/c =
0,1, 2,319, 2,31, 1,57, con una lmax = 10.

./poynting2D.pl -eps 10 -w 0.1 -lmax 10 -xmax 5 -N 100

./poynting2D.pl -eps 10 -w 2.319 -lmax 10 -xmax 5 -N 100

./poynting2D.pl -eps 10 -w 2.31 -lmax 10 -xmax 5 -N 100

./poynting2D.pl -eps 10 -w 1.57 -lmax 10 -xmax 5 -N 100

Figura 25: Ejecución del programa 24.

En la grá�ca 26 mostramos el campo magnético al cuadrado normalizado

rH2/H2
i con la distancia r y el campo Hi de la onda incidente, proporcio-

nal a distancias grandes a la sección diferencial de esparcimiento, para una

frecuencia pequeña ω = 0,1c/R y para polarización TM.

Figura 26: Cuadrado del campo H normalizado al campo incidente Hi y

escalado con r para polarización TM y para una frecuencia baja ω = 0,1.

Se ve claramente un patrón de campo cercano de caracter dipolar y a lo

lejos un patrón de radiación radiación dipolar con el dipolo orientado a lo

largo de la dirección vertical, que es la dirección del campo incidente que se

propaga hacia la derecha. Como estoy multiplicando el cuadrado del campo

por la distancia r, el valor asintótico sólo depende de la orientación.

En la grá�ca 27 la curva correspondiente para polarización TE. En este

Figura 27: Cuadrado del campo eléctrico, como la �gura 26 pero para pola-

rización TE.

caso el campo eléctrico apunta en la dirección del eje, induciendo un dipolo

axial que radía isotrópicamente alrededor. Como para esta polarización y en

el límite cuasiestático no hay campo de depolarización, el dipolo inducido y

la radiación son mayores.

La �gura 28 corresponde a una frecuencia relativamente alta ω = 2,319c/R
que es además una de las resonancias altas y angostas de las �guras 14 y 16.

Notamos claramente un patrón de radiación con 8 lóbulos, correspondiente

a una resonancia octopolar l = 4. Si disminuimos ligeramente la frecuencia,

por ejemplo, a ω = 2,31c/R, (�g. 29 la intensidad disminuye notablemente

y se vuelve más visible la dispersión frontal, responsable de extinguir al haz

incidente, de acuerdo al teorema óptico.

Algo similar sucede cerca de las resonancias para polarización TE, como

muestra la �gura 30 caculada a ω = 1,57c/R para polarización TE. Se ob-

Figura 28: rH2/H2
i , como en la �gura 26 para en una frecuencia ω =

2,319c/R cercana a una resonancia.

Figura 29: rH2, como en la �gura 28 para en una frecuencia ligeramente

menor ω = 2,31c/R.

Figura 30: rE2, como en la �gura 28 pero para polarización TE a una fre-

cuencia ω = 1,57c/R cercana a una resonancia.

serva claramente un patrón de 6 lóbulos correspondiente a una resonancia

hexapolar con l = 3.

7. Conclusiones

Desarrollamos la teoría para el cálculo del esparcimiento por cilindros

dieléctricos en términos de funciones de Bessel. Obtuvimos la ampli�cación

del campo en la super�cie, la cual puede ser un par de órdenes de magnitud

mayor que el campo incidente, y mostramos las resonancias tipo Mie y su

manifestación en las secciones transversales de dispersión.

8. Apéndice. Gaussiana imaginaria

En este apéndice mostramos como obtener la integral (31). Para ello,

consideremos primero la integral

I =

∫
C
dζeiζ2, (44)

sobre el circuito cerrado mostrado en la �g. 31. Como no hay singularidades

en su interior, la integral es I = 0. Por otro lado, la integral puede separarse
en tres partes como beginexport latex

0 = I =

∫ R

0
dζ eiζ

2
+ .

∫ π/4

0
d(Reiθ) ei(Re

iθ)2 −
∫ R

0
d(t
√
i)ei(t

√
i)2

=

∫ R

0
dζ eiζ

2
+ .

∫ π/4

0
dθ Reiθ eiR

2ei2θ − sqrti
∫ R

0
dt e−t

2
.

(45)

#+endexport El segundo término incluye un factor de la formaR exp(iR2(cos 2θ+
i sin 2θ) = R exp(iR2 cos 2θ) exp(−R2 sin 2θ). Para θ ∈ (0, π/4), sin 2θ > 0
y ese factor tiene magnitud menor a 1 y tiende a cero rápidamente cuando

R→∞. Por lo tanto, en el límite R→∞ podemos escribir beginexport latex∫ ∞
0

dζ eiζ
2
=
√
i

∫ ∞
0

dt e−t
2
=
sqrtiπ

2
, (46)

#+endexport de donde el resultado se sigue de inmediato.

9. Apéndice. Códigos de metapost

Código para producir las �guras 1 y 31.

Re ζ

Im ζ

0 R

Figura 31: Circuito C para realizar la integral (44) consistente en una recta

de 0 hasta R sobre el eje real, un octavo de círculo de radio R y el retorno

recto al origen.

g_use_svg = 0; %0 for postscript, 1 for svg

prologues:=3;

u=1cm;

outputtemplate := "%j-%c.eps";

if g_use_svg > 0:

outputtemplate := "%j-%c.svg";

outputformat := "svg";

fi

color yellow;

yellow=green+red;

beginfig(1)

path p[];

p0=fullcircle xscaled 3u yscaled 1u;

p1=p0 shifted (10u*up);

p2=subpath (-4,0) of p0--subpath(0,-4) of p1--cycle;

fill p2 withcolor .5blue+.5white;

fill p1 withcolor .7blue+.3white;

p3=unitsquare xscaled 10u yscaled 2u slanted -.5 rotated 90;

for i=-15 upto -10:

fill p3 shifted (i*((.5u,0))rotated -10) withcolor yellow;

draw p3 shifted (i*(.5u,0)rotated -10);

endfor;

drawarrow (((0,0)--(2u,0)) shifted (-10*(.5u,0))) rotated -10

shifted .5[point 0 of p3, point 2 of p3] withpen pencircle scaled

2pt;

label.top(btex $\vec q$ etex scaled 2, (1.5u,0) shifted (-10*(.5u,0))) rotated -10 shifted .5[point 0

of p3, point 2 of p3] withpen

pencircle scaled 2pt;

drawarrow (((0,0)--(2u,0)) shifted (10*(.5u,0))) rotated -10 shifted

.5[point 0 of p3, point 2 of p3] withpen pencircle scaled 2pt;

drawarrow (((0,0)--(2u,0)) shifted (10*(.5u,0))) rotated -30 shifted

.5[point 0 of p3, point 2 of p3] withpen pencircle scaled 2pt;

drawarrow (((0,0)--(2u,0)) shifted (10*(.5u,0))) rotated 10 shifted

.5[point 0 of p3, point 2 of p3] withpen pencircle scaled 2pt;

drawarrow ((0,0)--(0,1u)) shifted(0,-2u);

drawarrow ((0,0)--(1u,0)) rotated -10 shifted (0,-2u);

drawarrow ((0,0)--(0,-.75u)) slanted -.5 rotated 90 shifted (0,-2u);

label.ulft(btex $z$ etex scaled 2, (0,1u) shifted (0,-2u));

label.lrt(btex $x$ etex scaled 2,(1u,0) rotated -10 shifted (0,-2u));

label.urt(btex $y$ etex scaled 2, (0,-.75u) slanted -.5 rotated 90 shifted (0,-2u));

interim ahangle:=180;

drawdblarrow ((0,0)--(point 0 of p0))shifted (0,11u);

label.top(btex $R$ etex scaled 2, (point 0 of p0) scaled .5 shifted (0,11u));

endfig;

beginfig(2)

path p[];

p0=(0,0)--(5u,0)..subpath(0,1) of fullcircle scaled 10u--cycle;

drawdblarrow (0,7u)--(0,0)--(7u,0);

label.lrt(btex Re$\,\zeta$ etex scaled 1.5, (7u,0));

label.ulft(btex Im$\,\zeta$ etex scaled 1.5, (0,7u));

label.bot(btex $0$ etex scaled 1.5, (0,0));

label.bot(btex $R$ etex scaled 1.5, point 1 of p0);

drawarrow subpath(0,.5)of p0 withpen pencircle scaled 2pt;

drawarrow subpath(.5,2.5)of p0 withpen pencircle scaled 2pt;

drawarrow subpath(2.5,3.5)of p0 withpen pencircle scaled 2pt;

draw subpath(3.5,4)of p0 withpen pencircle scaled 2pt;

endfig;

end.

Este código se puede correr con el comando

mpost notas

